
Overview

The "Basic" part of Visual Basic refers to the BASIC (Beginners All-Purpose Symbolic Instruction Code) language, a language used by more programmers than any other language in the history of computing. 

Visual Basic has evolved from the original BASIC language and now contains several hundred statements, functions, and keywords, many of which relate directly to the Windows GUI.

This introduction will include the following topics:

1. Introduction to the Visual Basic Environment

2. Visual Basic Programming Fundamentals

3. Developing a Simple Visual Basic Application

4. Internet Programming: Visual Basic ActiveX Controls

5. Internet Programming: Server-Side Programming with the WebClass

Introduction to the Visual Basic Environment

The following discussion is reprinted from: http://www.msdn.microsoft.com/library/devprods/vs6/vbasic/vbcon98/vbconhellovisualbasic.htm
There are three main steps to creating an application in Visual Basic: 

1. Create the interface.

2. Set properties.

3. Write code. 

Creating the Interface

Forms are the foundation for creating the interface of an application. 

The first step in building a Visual Basic application is to create the forms that will be the basis for your application's interface.

Resizing, Moving, and Locking Controls

To draw a control using the Toolbox :

1. Click the tool for the control you choose to draw — in this case, the text box.


2. Move the pointer onto your form. The pointer becomes a cross hair, as shown in Figure 2.3. 

Figure 2.3 Drawing a text box with the Toolbox

PRIVATE "TYPE=PICT;ALT="
3. Place the cross hair where you want the upper-left corner of the control.


4. Drag the cross hair until the control is the size you want. (Dragging means holding the left mouse button down while you move an object with the mouse.)


5. Release the mouse button.  The control appears on the form.

Resizing, Moving, and Locking Controls

To resize a control :

1. Select the control you intend to resize by clicking it with the mouse. 

Sizing handles appear on the control.

2. Position the mouse pointer on a sizing handle, and drag it until the control is the size you choose. 

The corner handles resize controls horizontally and vertically, while the side handles resize in only one direction.

3. Release the mouse button. 

–or–

Use SHIFT with the arrow keys to resize the selected control.

To move a control :

· Use the mouse to drag the control to a new location on the form. 

–or–

Use the Properties window to change the Top and Left properties.

To lock all control positions:

· From the Format menu, choose Lock Controls. 

–or–

Click the Lock Controls Toggle button on the Form Editor toolbar.

Figure 2.4 The interface for the "Hello, world!" application

Setting Properties

The next step is to set properties for the objects you've created. The Properties window (Figure 2.5) provides an easy way to set properties for all objects on a form

PRIVATE "TYPE=PICT;ALT="The Properties window consists of the following elements: 

· Object box — Displays the name of the object for which you can set properties. Click the arrow to the right of the object box to display the list of objects for the current form.


· Sort tabs — Choose between an alphabetic listing of properties or a hierarchical view divided by logical categories, such as those dealing with appearance, fonts, or position. 


· Properties list — The left column displays all of the properties for the selected object. You can edit and view settings in the right column. 

Figure 2.5 The Properties window
To set properties from the Properties window:

1. From the View menu, choose Properties, or click the Properties button on the toolbar. 

The Properties window displays the settings for the selected form or control.

2. From the Properties list, select the name of a property.


3. In the right column, type or select the new property setting. 

Enumerated properties have a predefined list of settings. You can display the list by clicking the down arrow at the right of the Settings box, or you can cycle through the list by double-clicking a list item.

For the "Hello, world!" example, you'll need to change three property settings:

	PRIVATE
Object
	Property
	Setting

	Form
	Caption
	Hello, world!

	Text box
	Text
	(Empty)

	Command button
	Caption
	OK


Writing Code

The Code Editor window is where you write Visual Basic code for your application. 

Code consists of language statements, constants, and declarations. PRIVATE "TYPE=PICT;ALT="
The Code window includes the following elements: 

· Object list box — Displays the name of the selected object. Click the arrow to the right of the list box to display a list of all objects associated with the form.


· Procedure list box — Lists the procedures, or events, for an object. The box displays the name of the selected procedure — in this case, Click. Choose the arrow to the right of the box to display all the procedures for the object. 

To open the Code window:

· Double-click the form or control for which you choose to write code. 

–or–

From the Project Explorer window, select the name of a form or module, and choose the View Code button.

Figure 2.6 The Code Editor window

Creating Event Procedures

Code in a Visual Basic application is divided into smaller blocks called procedures. 

An event procedure, such as those you'll create here, contains code that is executed when an event occurs (such as when a user clicks a button). 

An event procedure for a control combines the control's actual name (specified in the Name property), an underscore (_), and the event name. 

To create an event procedure:

1. In the Object list box, select the name of an object in the active form. (The active form is the form that currently has the focus.) 

For this example, choose the command button, Command1.

2. In the Procedure list box, select the name of an event for the selected object. 

Here, the Click procedure is already selected, because it's the default procedure for a command button. Note that a template for the event procedure is now displayed in the Code window.

3. Type the following code between the Sub and End Sub statements: 

Text1.Text = "Hello, world!"

The event procedure should look like this:

Private Sub Command1_Click ()

   Text1.Text = "Hello, world!"

End Sub

Running the Application

To run the application, choose Start from the Run menu, or click the Start button on the toolbar, or pressF5. 

Visual Basic Programming Fundamentals

The following discussion is reprinted from:

http://www.msdn.microsoft.com/library/devprods/vs6/vbasic/vbcon98/vbconcodebasics.htm
Declaring Variables

To declare a variable is to tell the program about it in advance. You declare a variable with the Dim statement, supplying a name for the variable:

Dim variablename [As type]
Implicit Declaration

You don't have to declare a variable before using it. For example, you could write a function where you don't need to declare TempVal before using it:

Function SafeSqr(num)

   TempVal = Abs(num)

   SafeSqr = Sqr(TempVal)

End Function

Explicit Declaration

To avoid the problem of misnaming variables, you can stipulate that Visual Basic always warn you whenever it encounters a name not declared explicitly as a variable.

To explicitly declare variables:

· Place this statement in the Declarations section of a class, form, or standard module: 

Option Explicit
Creating Your Own Constants

The syntax for declaring a constant is:

[Public|Private] Const constantname[As type] = expression
Declaring Variables with Data Types

Before using a non-Variant variable, you must use the Private, Public, Dim or Static statement to declare it As type. For example, the following statements declare an Integer, Double, String, and Currency type, respectively:

Private I As Integer

Dim Amt As Double

Static YourName As String

Public BillsPaid As Currency

A Declaration statement can combine multiple declarations, as in these statements:

Private I As Integer, Amt As Double

Private YourName As String, BillsPaid As Currency

Private Test, Amount, J As Integer

Data types supported by Visual Basic include:

· Integer,

· Long (long integer)

· Single (single-precision floating point)

· Double (double-precision floating point)

· Currency

· Byte

· String

· Boolean

· Date

· Variant

Sub Procedures

A Sub procedure is a block of code that is executed in response to an event. 

By breaking the code in a module into Sub procedures, it becomes much easier to find or modify the code in your application.

The syntax for a Sub procedure is:

[Private|Public][Static]Sub procedurename (arguments)
statements
End Sub
There are two ways to call a Sub procedure:

1. Call MyProc (FirstArgument, SecondArgument)

2. MyProc FirstArgument, SecondArgument

Function Procedures

Visual Basic includes built-in, or intrinsic functions, like Sqr, Cos or Chr. In addition, you can use the Function statement to write your own Function procedures.

The syntax for a Function procedure is:

[Private|Public][Static]Function procedurename (arguments) [As type]
statements
End Function
To call a function:

X = AnotherFunction(10 * ToDec)

Passing Arguments By Value

Only a copy of a variable is passed when an argument is passed by value. 

If the procedure changes the value, the change affects only the copy and not the variable itself. Use the 

For example:

Sub PostAccounts(ByVal intAcctNum as Integer)

   .

   . ' Place statements here.

   .

End Sub

Passing Arguments By Reference

Passing arguments by reference gives the procedure access to the actual variable contents in its memory address location. 

If you specify a data type for an argument passed by reference, you must pass a value of that type for the argument. You can work around this by passing an expression, rather than a data type, for an argument. Visual Basic evaluates an expression and passes it as the required type if it can.

This is the default mode for visual basic procedures and functions.

If...Then

Use an If...Then structure to execute one or more statements conditionally:

If condition Then statement
If condition Then
statements
End If
If...Then...Else

Use an If...Then...Else block to define several blocks of statements, one of which will execute:

If condition1 Then
[statementblock-1]
[ElseIf condition2 Then
[statementblock-2]] ...
[Else
[statementblock-n]]
End If
Select Case

Visual Basic provides the Select Case structure as an alternative to If...Then...Else for selectively executing one block of statements from among multiple blocks of statements. 

A Select Case structure works with a single test expression that is evaluated once, at the top of the structure. Visual Basic then compares the result of this expression with the values for each Case in the structure. If there is a match, it executes the block of statements associated with that Case:

Select Case testexpression
[Case expressionlist1
[statementblock-1]]
[Case expressionlist2
[statementblock-2]]
.
.
.
[Case Else
[statementblock-n]]
End Select
Do...Loop

Use a Do loop to execute a block of statements an indefinite number of times. 

As with If...Then, the condition must be a value or expression that evaluates to False (zero) or to True (nonzero).

In the following Do...Loop, the statements execute as long as the condition is True:

Do While condition
statements
Loop
For...Next

When you know you must execute the statements a specific number of times, however, a For…Next loop is a better choice than the Do loop.

 Unlike a Do loop, a For loop uses a variable called a counter that increases or decreases in value during each repetition of the loop. The syntax is:

For counter = start To end [Step increment]
statements
Next [counter]
For Each...Next

A For Each...Next loop is similar to a For...Next loop, but it repeats a group of statements for each element in a collection of objects or in an array instead of repeating the statements a specified number of times. 

Here is the syntax for the For Each...Next loop:

For Each element In group
statements
Next element
Exiting a Control Structure

The Exit statement allows you to exit directly from a For loop, Do loop, Sub procedure, or Function procedure. 

The syntax for the Exit statement is simple: Exit For can appear as many times as needed inside a For loop, and Exit Do can appear as many times as needed inside a Do loop:

For counter = start To end [Step increment]
[statementblock]
[Exit For]
[statementblock]
Next [counter[, counter] [,...]]
Developing a Simple Visual Basic Application: HtmlViewer

Again, follow the three main steps to creating an application in Visual Basic: 

1. Create the interface.

2. Set properties.

3. Write code. 

The HtmlViewer Interface

The HtmlViewer interface will appear as follows:

Setting Interface Properties

“Main” Form: 

· The caption property is set to “Html Source Viewer”.

· The border style is set to “fixed single”.  This will prevent the main window from being resized.

Control Labels:

· The caption for each label is modified to describe its associated control.

“DirectoryBox” Textbox:

· Clear the text Field.

“Search” Command Button:

· Set the button caption to “Search”.

“Selected FilesBox” Textbox:

· Clear the text field.

Writing Code

Two Visual Basic units are used for this program: Main.frm and InetFile.bas.

Main.frm:

' Attribute:   Files

' Description: A set of html files found in the directory tree

'              starting at the specified path.

Private Files As Collection

' Function:    Form_Load

' Description: Method used to initialize the form. The file

'              collection will be created, and the directory textbox

'              will be initialized to the default html directory.

Private Sub Form_Load()

    'Create a new collection.

    Set Files = New Collection

    'Initialize the directory textbox.

    DirectoryBox.Text = GetInternetDirectory

End Sub

' Function:    Form_Terminate

' Description: Method used to unload the form. The file

'              collection will be deallocated.

Private Sub Form_Terminate()

    'Free the file collection.

    Set Files = Nothing

End Sub

' Function:    SearchBtn_Click

' Description: This method will be called when the search button

'              is clicked.  The search path will be read from the

'              directory box, and the method will scan for html

'              files.  The html files found will be copied to the

'              file box for display.

Private Sub SearchBtn_Click()

    'Scan for html files at the specified path.

    Scan DirectoryBox.Text, Files

    'Copy the html files found at the path to the file box and

    'remove them from the list.

    While Files.Count > 0

       FileBox.AddItem Files(1)

       Files.Remove (1)

    Wend

End Sub

' Function:    FileBox_Click

' Description: Load the selected file to the detail box.

Private Sub FileBox_Click()

    'Load the selected file to the file detail box.

    FileDetailBox.LoadFile (FileBox.List(FileBox.ListIndex))

End Sub

InetFile:

' Function:    GetInternetDirectory

' Description: Retreive the default directory used by the system to

'              cache html files.

Function GetInternetDirectory() As String

    'Create a local file system object to create a path.

    Set fs = CreateObject("Scripting.FileSystemObject")

    Set WinFolder = fs.GetSpecialFolder(WindowsFolder)

    'Construct a html path from the windows path.

    GetInternetDirectory = WinFolder.Path & "\Temporary Internet Files"

    'Free the file system objects.

    Set WinFolder = Nothing

    Set fs = Nothing

End Function

' Function:    Scan

' Description: Scan the specified path for html files. Return the set

'              of html files found in the path, including subdirectories,

'              as a collection. Subdirectories will be searched by

'              recursively calling this routine.

Sub Scan(Dir As String, _

         Results As Collection)

    'Create a local file system object to search for HTML files.

    Set fs = CreateObject("Scripting.FileSystemObject")

    'Get the root folder

    Set ffolder = fs.GetFolder(Dir)

    'Get the subfolders for the root folder.

    Set sfolder = ffolder.SubFolders

    'Recursively scan all subfolders in the root. Handle any

    'new events between recursive calls.

    For Each fitem In sfolder

        Scan fitem.Path, Results

        DoEvents

    Next

    'Get the files for the root folder.

    Set allfiles = ffolder.Files

    'For each file, append it to a collection.

    For Each fileitem In allfiles

        If (fileitem.Type = "HTML Document") Or _

           (fileitem.Type = "Microsoft HTML Document 4.0") Or _

           (fileitem.Type = "Netscape Hypertext Document") Then

            Results.Add fileitem.Path

        End If

    Next

    'Free the file system objects.

    Set allfiles = Nothing

    Set sfolder = Nothing

    Set ffolder = Nothing

    Set fs = Nothing

End Sub
Internet Programming: Scripting

Available on both the client-side and server side.

Uses a subset of the Visual Basic language.

Details will be left for Lecture 5.

Internet Programming: Visual Basic ActiveX Controls

An ActiveX control is a COM component.  COM, or Component Object Model, is a Microsoft standard for creating interchangeable software components.

ActiveX controls are installed on the client or server, and interact within the browser/server object model.

On the client, the component is embedded in an HTML page.  When the page is downloaded from the server, the control is also downloaded and installed on the client.

The HtmlViewer program can be converted into a control by performing the following steps:

1. Create a new ActiveX Control project.

2. Copy the controls from the HtmlViewer Main form into the user control.

3. Copy the code from the HtmlViewer Main form into the user control.  Change references to the Main form within the code to the new control name.

4. Add the InetFile.bas file to the control project.

5. Recompile and register the control with the system register (done automatically on the development machine).

The control can now be embedded into an HTML page:

<p /><object id="HtmlViewer"

classid="clsid:70BD2F03-504F-11D5-B23A-00A0C96FE39E"

align="baseline" border="0" width="500" height="350"

alighn="center"></object> </p>

</body>

</html>
Internet Programming: Server-Side Programming with the WebClass

Like the ActiveX control, the WebClass is a COM component.  It is installed on the web server.

The WebClass dynamically generates HTML pages.  It does so through two mechanisms: the Response.Write method and the WebItem.WriteTemplate.

WebItems are HTML templates that are used by the WebClass.  There are two kinds of WebItems: HTML template WebItems and Custom WebItems.

The WebClass is event driven, starting with the WebClass_Start() event.  An initial HTML page is generated in response to the Start event.  Additional  HTML pages may be generated in response to other events.

The file system scan portion of the HtmlViewer can be implemented as a WebClass by:

1. Create a request HTML page to submit a query.  The SubmitRequest HTML source appears as:

<html>

<body>

   <form method=post>

      <input Name="Path">

      <input type=submit value="Search"> 

   </form>

</body>

</html>

2. Create a results HTML page to display the results of the query.  The Results HTML source appears as:

<html>

<body>

Files found at <WC@PATH> path </WC@PATH>: 

<WC@RESULTS> <br>No html files were found. </WC@RESULTS>

</body>

</html>

3. Add the SubmitRequest and Results files to the WebClass as WebItems.

4. Add a Submit event to the SubmitRequest WebItem.  The Submit event will perform the same operation as the command button of the HtmlViewer.  Instead of writing the results to a list box, the Submit event will use the Results WebItem to output the query results. 

5. Add the ProcessTags event to the Results WebItem.  This event will be called when the Results HTML page is generated.  It will replace the WC@PATH tag with the query path, and the WC@RESULTS tag with the HTML files found.

The remainder of the WebClass events are similar to those of the HtmlViewer.

The SubmitRequest page is presented as the default page by the WebClass:

The Results page is presented after the WebClass has scanned for HTML files on the server:

Conclusions

Visual Basic is an effective tool for creating applications or composite components out of lower level ActiveX/COM components.

Having evolved from BASIC, Visual Basic has a “relaxed” syntax and structure.  Programming style should not be compromised as a result of this shortcoming.

